If this is your first visit, be sure to check out the FAQ by clicking the link above.
You may have to register before you can post: click the register link above to proceed.
To start viewing messages, select the forum that you want to visit from the selection below.

I have several spreadsheets which I use for data gathering (usually in
columns, but could be in rows) and where I need to do calculations
based on the last (ie most recent, at the bottom) entry in the column.

Although Excel has functions designed to help me find the largest,
smallest, average etc etc of these numbers, I have yet to find one
which returns the last entry. Of course, it is not hard to add an extra
(subsequently hidden) column which does the calculations (using the if
and isblank functions) but I'm sure there must be a better way to do it
- with a formula in a single cell.

Depending on what you consider to be the "last" item, see if one of these
does what you want:

The value of the last non-blank cell in ROW range
=LOOKUP(2,1/(A1:M1<>""),A1:M1)

The value of the last non-blank cell in a COLUMN range
=LOOKUP(2,1/(A1:A100<>""),A1:A100)

The VALUE of the last numeric value in Col_A:
=LOOKUP(10^99,A:A)

The VALUE of the last numeric value in Row_1:
=LOOKUP(10^99,1:1)

The VALUE of the last text cell in Col_A
=LOOKUP(REPT("z",255),A:A)

The VALUE of the last text cell in Row_1
=LOOKUP(REPT("z",255),1:1)

Does that help?
***********
Regards,
Ron

XL2002, WinXP

"Victor Delta" wrote:
> I have several spreadsheets which I use for data gathering (usually in
> columns, but could be in rows) and where I need to do calculations
> based on the last (ie most recent, at the bottom) entry in the column.
>
> Although Excel has functions designed to help me find the largest,
> smallest, average etc etc of these numbers, I have yet to find one
> which returns the last entry. Of course, it is not hard to add an extra
> (subsequently hidden) column which does the calculations (using the if
> and isblank functions) but I'm sure there must be a better way to do it
> - with a formula in a single cell.
>
> Any suggestions please?
>
> Thanks
>
> V
>
>